
' ACQUISITION ET ANALYSE DES SIGNAUX

' ©Gérard Pinson

'

' Ce programme permet de :

'- faire l'acquisition d'un signal échantillonné (quantifié sur 8 bits) par le port série

'- ou générer un signal échantillonné en mode simulation

'- calculer le spectre d'amplitude du signal

'- afficher le graphe du signal et/ou du spectre

'- imprimer les valeurs numériques des échantillons du signal et/ou du spectre

'NB : ce programme est également utilisable pour afficher un graphe quelconque de M courbes de N points

'les valeurs des abscisses sont stockées dans le tableau à 1 dimension x(n)

'les valeurs des ordonnées sont stockées dans le tableau à 2 dimensions y(n,m)

' Définition des variables structurées nécessaires aux graphiques

TYPE AXE

    min AS SINGLE: 'borne inférieure (calculée par défaut si = 0)

    max AS SINGLE: 'borne supérieure (calculée par défaut si = 0)

    unit AS SINGLE: 'unité (calculée par défaut si = 0)

    ncs AS INTEGER: 'nombre de chiffres significatifs des graduations (valeur par défaut : 3)

    linlog AS INTEGER: 'type d'axe : linéaire = 0 (défaut) ; logarithmique = -1

END TYPE

TYPE GRAPHE

    mode AS INTEGER: ' mode graphique

    grille AS INTEGER: 'existence (=-1) ou non (=0 ; valeur par défaut) d'une grille

    ligne AS INTEGER: 'allure du tracé : rien = 0 (défaut), lignes = 1 , barres = 2,...)

    forme AS INTEGER: 'forme des points : point = 0 (défaut), croix = 1, cercle = 2,...)

END TYPE

' Procédures principales

DECLARE SUB Commin (x() AS SINGLE, y() AS SINGLE)

DECLARE SUB gbf (x() AS SINGLE, y() AS SINGLE)

DECLARE SUB graph (x() AS SINGLE, Xax AS AXE, y() AS SINGLE, Yax AS ANY, aspect AS GRAPHE)

DECLARE SUB spectreampli (x() AS SINGLE, y() AS SINGLE)

' Procédures auxiliaires

DECLARE FUNCTION affichegrad$ (s AS SINGLE, Sax AS AXE)

DECLARE FUNCTION wlog! (s AS SINGLE, flaglinlog AS INTEGER)

DECLARE SUB alerte (message AS STRING)

DECLARE SUB calculgrad (Sax AS AXE, sgrad() AS SINGLE)

DECLARE SUB CalculXlin (x() AS SINGLE, xmin AS SINGLE, xmax AS SINGLE)

DECLARE SUB CalculExtrema (y() AS SINGLE, yminf AS SINGLE, ymaxf AS SINGLE, indexmin AS INTEGER, indexmax AS INTEGER, yposmin AS SINGLE)

DECLARE SUB FFT (PartRe() AS SINGLE, PartIm() AS SINGLE, choix AS STRING)

DECLARE SUB imprim (x() AS SINGLE, y() AS SINGLE)

' Définitions diverses

ON ERROR GOTO 0

CONST formatreel = 14:  'nb de caractères d'un réel en simple précision

CONST pi# = 3.141592653589795#

' Définition du port de communication série

'format : 1200 bauds, sans parité, 8 bits, 1 stop

DIM SHARED ComSerie AS STRING

ComSerie = "COM2: 1200, N, 8, 1"

DEFINT E-O: 'définitions par défaut :

            '-variable commenáant par une lettre comprise entre E et O : INTEGER

            '-autres variables : SINGLE, sauf spécification contraire

DIM aspect AS GRAPHE

aspect.mode = 12:  'ecran VGA 16 couleurs

DIM Xax AS AXE

DIM Yax AS AXE

DIM defaut AS AXE

DIM RESP AS STRING

defaut.min = 0: 'valeurs par défaut des paramètres de définition d'un axe

defaut.max = 0

defaut.unit = 0

defaut.linlog = 0

defaut.ncs = 3

CLS

DEBUT:

CLS

' Définition des tableaux de valeurs

INPUT "nb de points (puissance de 2)"; N1

n = N1 - 1

m = 0

REDIM x(n) AS SINGLE, y(n, m) AS SINGLE

REM placer ici vos instructions **************************************

'l'appel des différentes procédures est laissé au choix de l'utilisateur

'ce qui suit est un exemple (voir TP B24)

'

'affichage d'un signal par GBF puis de son spectre

'simulation du signal

gbf x(), y()

'affichage

Xax = defaut

Yax = defaut

aspect.grille = 0

aspect.ligne = 3

graph x(), Xax, y(), Yax, aspect

'analyse de Fourier

spectreampli x(), y()

'affichage du spectre

Xax.min = 0

Xax.max = N1 / 2

Xax.unit = N1 / 16

Yax = defaut

aspect.grille = -1

aspect.ligne = 2

graph x(), Xax, y(), Yax, aspect

REM *****************************************************************

SCREEN 0

END

DEFSTR A-D

FUNCTION affichegrad$ (s AS SINGLE, Sax AS AXE)

'fonction servant à l'affichage des graduations

'valeur par défaut

IF Sax.ncs = 0 THEN Sax.ncs = 3

IF Sax.linlog THEN GOSUB affichegradlog ELSE GOSUB affichegradlin

EXIT FUNCTION

affichegradlin:

's = nb à arrondir

IF ABS(s) < Sax.unit / 10 ^ Sax.ncs THEN

    affichegrad$ = " 0"

    RETURN

END IF

affichegradarrondie:

ss# = CDBL(s)

spos# = ABS(ss#)

scale# = 10 ^ INT(LOG(spos#) / LOG(10))

sech# = spos# / scale#

smult# = sech# * 10 ^ (Sax.ncs - 1)

smint# = INT(smult#)

IF smult# - smint# <= .5 THEN smar# = smint# ELSE smar# = 1 + smint#

sechar# = smar# / 10 ^ (Sax.ncs - 1)

sarabs# = sechar# * scale#

sar# = SGN(s) * sarabs#

affichegrad$ = STR$(CSNG(sar#))

RETURN

affichegradlog:

slog = LOG(s) / LOG(10)

slogint = INT(slog)

labtestlog = INT(slogint - slog)

IF NOT labtestlog THEN GOSUB affichegradarrondie

RETURN

END FUNCTION

DEFSNG A-D

SUB alerte (message AS STRING)

    BEEP

    PRINT message

    PRINT "appuyer sur une touche pour continuer"

    DO: LOOP WHILE INKEY$ = ""

END SUB

DEFSTR A-D

SUB CalculExtrema (y() AS SINGLE, yminf AS SINGLE, ymaxf AS SINGLE, indexmin AS INTEGER, indexmax AS INTEGER, yposmin AS SINGLE)

'calcule les extrema du tableau y(i,j)

'détermine l'indice i de ces valeurs

n = UBOUND(y, 1)

m = UBOUND(y, 2)

indexmin = 0

indexmax = 0

yminf = y(0, 0)

ymaxf = y(0, 0)

yposmin = 3.402823E+38

FOR J = 0 TO m

    FOR I = n TO 0 STEP -1

        IF y(I, J) < yminf THEN yminf = y(I, J): indexmin = I

        IF y(I, J) > ymaxf THEN ymaxf = y(I, J): indexmax = I

        IF y(I, J) > 0 AND y(I, J) < yposmin THEN yposmin = y(I, J)

    NEXT I

NEXT J

END SUB

DEFSNG A-D

SUB calculgrad (Sax AS AXE, sgrad() AS SINGLE)

IF Sax.unit = 0 THEN

    Rs = 10 ^ INT(LOG(Sax.max - Sax.min) / LOG(10))

    Ts = (Sax.max - Sax.min) / Rs

    IF Ts <= 2 THEN

        Sax.unit = .2

    ELSEIF Ts > 5 THEN

        Sax.unit = 1

    ELSE

        Sax.unit = .5

    END IF

    Sax.unit = Sax.unit * Rs

END IF

IF Sax.linlog THEN GOSUB calculgradlog ELSE GOSUB calculgradlin

EXIT SUB

calculgradlin:

'détermination du nb de graduations

IF Sax.min < 0 AND Sax.max >= 0 THEN sorig = 0 ELSE sorig = Sax.min

s1 = (sorig - Sax.min) / Sax.unit

s2 = INT(s1)

sdelta = (s1 - s2) * Sax.unit

I = 0

WHILE s <= Sax.max

    s = I * Sax.unit + Sax.min + sdelta

    I = I + 1

WEND

Nsgrad = I - 2

ERASE sgrad

REDIM sgrad(Nsgrad) AS SINGLE

FOR I = 0 TO Nsgrad

sgrad(I) = I * Sax.unit + Sax.min + sdelta

NEXT I

RETURN

calculgradlog:

sdec = 10 ^ INT(LOG(Sax.min) / LOG(10))

puissance = 1

s = sdec

K = 0

WHILE s <= Sax.max

    s = sdec * puissance

    I = 1

    WHILE s < Sax.min

        I = I + 1

        s = I * sdec * puissance

    WEND

    WHILE s < Sax.max AND I <= 10

        s = I * sdec * puissance

        I = I + 1

        K = K + 1

    WEND

    puissance = puissance * 10

WEND

Nsgrad = K - 1

ERASE sgrad

REDIM sgrad(Nsgrad) AS SINGLE

puissance = 1

s = sdec

K = 0

WHILE s <= Sax.max

    s = sdec * puissance

    I = 1

    WHILE s < Sax.min

        I = I + 1

        s = I * sdec * puissance

    WEND

    WHILE s < Sax.max AND I <= 10

        s = I * sdec * puissance

        sgrad(K) = s

        I = I + 1

        K = K + 1

    WEND

    puissance = puissance * 10

WEND

RETURN

END SUB

DEFSTR A-D

SUB CalculXlin (x() AS SINGLE, xmin AS SINGLE, xmax AS SINGLE)

n = UBOUND(x)

xl = xmax - xmin

FOR I = 0 TO n

    x(I) = xl * I / n + xmin

NEXT I

END SUB

SUB Commin (x() AS SINGLE, y() AS SINGLE)

'SUB de réception d'un fichier binaire (suite d'octets)

'    si nécessaire, adapter les paramètres de communication (voir ci-dessous)

'Le fichier reáu est enregistré dans le tableau y(i,0) de longueur N+1

'    si la longueur du fichier reáu est inférieure à N+1, des 0 sont inscrits dans la partie restante du tableau y(i,0)

'    si la longueur du fichier reáu est supérieure à N+1, les octets suivant le N+1ème sont perdus...

'Après un délai d'attente de une seconde,

'    s'il n'y a plus d'octets présents en entrée sur le port COM,

'    retour vers le programme principal.

'Un tableau x(i) de valeurs en abscisse est généré

n = UBOUND(y, 1)

xbornesup = n

CalculXlin x(), 0, xbornesup

PRINT "Récepteur pràt : lancer l'émission !"

OPEN ComSerie FOR RANDOM AS #1 LEN = 4096

entreemodem$ = ""

temPause = 1

DO

'    BEEP

    DO

        'Contrìle du modem. LOC(1) fournit le nombre de caractères en attente

        entreemodem$ = entreemodem$ + INPUT$(1, #1)

    LOOP WHILE LOC(1) > 0

    plusdedonnee = TIMER

    WHILE LOC(1) = 0

        IF TIMER - plusdedonnee > temPause THEN EXIT DO

    WEND

LOOP

findereception:

CLOSE #1

PRINT "Fin de réception"

I = 0

WHILE I <= n AND I < LEN(entreemodem$)

    y(I, 0) = ASC(MID$(entreemodem$, I + 1, 1))

    I = I + 1

WEND

END SUB

SUB FFT (PartRe() AS SINGLE, PartIm() AS SINGLE, choix AS STRING)

' * calcul de la transformation de Fourier rapide (FFT) : choix = "F"

'   ou de son inverse (IFT) : choix = "I"

'   par l'algorithme "papillon"

' * tableaux en OPTION BASE 0 (premier indice = 0)

' * LA DIMENSION DES TABLEAUX (+1) DOIT ETRE UNE PUISSANCE DE DEUX

LOCATE 2, 55

PRINT "calcul FFT en cours"

n = UBOUND(PartRe) + 1

m = LOG(n) / LOG(2)

DIM StockRe(n - 1) AS SINGLE, StockIm(n - 1) AS SINGLE

DIM TFRE(n / 2 - 1) AS DOUBLE, TFIM(n / 2 - 1) AS DOUBLE

IF choix = "I" THEN Signe = 1

IF choix = "F" THEN Signe = -1

IF Signe = -1 THEN GOTO ETIQ1

' réordonner les données pour IFT (transformée inverse)

FOR I = 0 TO n / 2

    StockRe(I) = PartRe(I + (n / 2 - 1))

    StockIm(I) = PartIm(I + (n / 2 - 1))

NEXT I

FOR I = 0 TO n / 2 - 2

    StockRe(I + (n / 2 + 1)) = PartRe(I)

    StockIm(I + (n / 2 + 1)) = PartIm(I)

NEXT I

FOR I = 0 TO n - 1

    PartRe(I) = StockRe(I)

    PartIm(I) = StockIm(I)

NEXT I

ETIQ1:

' calculer les Wnk, coefficients multiplicatifs complexes

pi2# = 2 * pi#

FOR O = 0 TO n / 2 - 1

    TFRE(O) = COS(pi2# * (-O) / n)

    TFIM(O) = Signe * SIN(pi2# * (-O) / n)

NEXT O

' calculer la FFT

FOR I = 1 TO m

    l = 0: h = 0

    G = n / 2 ^ I

    FOR K = 0 TO (n - 1) STEP G

        TFI = 0

        TFIFLAG = (-1) ^ (l + 1)

        FOR J = 0 TO (G - 1)

            TFI = J * 2 ^ (I - 1)

            R% = K + J: s% = J + h: T% = J + G + h

            IF TFIFLAG > 0 THEN GOTO ETIQ2

            StockRe(R%) = PartRe(s%) + PartRe(T%)

            StockIm(R%) = PartIm(s%) + PartIm(T%)

            GOTO ETIQ3

ETIQ2:

            TEMPRE = PartRe(s%) - PartRe(T%)

            TEMPIM = PartIm(s%) - PartIm(T%)

            StockRe(R%) = TEMPRE * TFRE(TFI) - TEMPIM * TFIM(TFI)

            StockIm(R%) = TEMPRE * TFIM(TFI) + TEMPIM * TFRE(TFI)

ETIQ3:

        NEXT J

    l = l + 1: h = INT(l / 2) * G * 2

    NEXT K

    FOR ii = 0 TO n - 1

        PartRe(ii) = StockRe(ii)

        PartIm(ii) = StockIm(ii)

    NEXT ii

NEXT I

FOR I = 0 TO n - 1

    PartRe(I) = PartRe(I) / n

    PartIm(I) = PartIm(I) / n

NEXT I

' renversement binaire pour restaurer les indices

FOR I = 0 TO n - 1

    INDEX% = I

    IOUT% = 0

    FOR J = 1 TO m

        TEMP% = 1 AND INDEX%

        IOUT% = IOUT% * 2

        IOUT% = IOUT% + TEMP%

        INDEX% = INDEX% \ 2

    NEXT J

    StockRe(I) = PartRe(IOUT%)

    StockIm(I) = PartIm(IOUT%)

NEXT I

IF Signe = 1 THEN GOTO ETIQ4

' ordonner les tableaux de sortie FFT

' fréquences négatives de 0 à N/2-2

' composante continue DC à N/2-1

' fréquences positives de N/2 à N-1

FOR I = 0 TO n / 2

    PartRe(I + (n / 2 - 1)) = StockRe(I)

    PartIm(I + (n / 2 - 1)) = StockIm(I)

NEXT I

FOR I = 0 TO n / 2 - 2

    PartRe(I) = StockRe(I + (n / 2 + 1))

    PartIm(I) = StockIm(I + (n / 2 + 1))

NEXT I

' La sortie FFT est ici, dans les tableaux PartRe pour les parties réelles

' et PartIm pour les parties imaginaires

EXIT SUB

' Mise à l'échelle de la sortie IFT en multipliant par N

ETIQ4:

FOR I = 0 TO n - 1

    PartRe(I) = StockRe(I) * n

    PartIm(I) = StockIm(I) * n

NEXT I

' La sortie IFT est ici, dans les tableaux PartRe pour les parties réelles

' et PartIm pour les parties imaginaires

END SUB

SUB gbf (x() AS SINGLE, y() AS SINGLE)

'Calcule un signal de y(i,0) de durée Tf (durée de la "fenàtre" temporelle)

DIM Vampli(10) AS SINGLE, Phi(10) AS SINGLE

n = UBOUND(y, 1)

N1 = n + 1

'choix de la forme du signal

PRINT "carré :         entrer 1"

PRINT "dent de scie :  entrer 2"

PRINT "sinus :         entrer 3"

PRINT "sinusred :      entrer 4"

PRINT "créneau :       entrer 5"

PRINT "Mod. MLI (PWM): entrer 6"

PRINT "Mod. amplitude: entrer 7"

PRINT "Mod. fréquence: entrer 8"

PRINT "Mod. FSK :      entrer 9"

PRINT "Dirac :         entrer 10"

PRINT "UN (DC) :       entrer 11"

PRINT "Synthèse de F.: entrer 12"

PRINT "SIN ech. :      entrer 13"

PRINT "SIN ech./bloq.: entrer 14"

entreeformesignal:

INPUT "type de signal"; formesignal

IF formesignal < 1 OR formesignal > 14 THEN

    BEEP

    GOTO entreeformesignal

END IF

PRINT

'définition des paramètres du signal

SELECT CASE formesignal

    CASE 1

        INPUT "amplitude"; v

        INPUT "rapport cyclique"; rapportcyclique

        INPUT "nb de périodes dans la fenàtre d'échantillonnage"; q

    CASE 5

        INPUT "amplitude"; v

        INPUT "angle d'amoráage (en degrés) (0 << 90)"; psi

        INPUT "nb de périodes dans la fenàtre d'échantillonnage"; q

    CASE 2, 3, 4, 10, 11

        INPUT "amplitude"; v

        INPUT "nb de périodes dans la fenàtre d'échantillonnage"; q

    CASE 6, 7, 8, 9

        PRINT "amplitude de la porteuse = 1"

        INPUT "amplitude du signal modulant"; v

        PRINT "fréquence du signal modulant = 1"

        INPUT "fréquence de la porteuse"; q

    CASE 12

        CLS

        PRINT "Entrez l'amplitude (en volts) et la phase (degrés) des 10 premiers harmoniques"

        PRINT "ainsi que, s'il y a lieu, la valeur (en volts) de la composante continue"

        PRINT

        INPUT "Composante continue"; Vampli(0)

        Phi(0) = 0

        PRINT

        PRINT "Entrez les valeurs amplitude, phase, séparées par une virgule:"

        PRINT "Amplitude, phase du fondamental"

        INPUT Vampli(1), Phi(1)

        Phi(1) = Phi(1) * pi# / 180

        FOR h = 2 TO 10

            PRINT "Harmonique de rang"; h

            INPUT Vampli(h), Phi(h)

            Phi(h) = Phi(h) * pi# / 180

        NEXT h

        q = 1

    CASE 13, 14

        INPUT "amplitude du signal"; v

        PRINT "fréquence du signal = 1"

        INPUT "fréquence d'échantillonnage"; q

END SELECT

Tf = 1:  'durée de l'observation du signal (fenàtre temporelle)

Ts = Tf / q:  'période du signal

Te = Tf / N1:  'periode d'échantillonnage

tpsi = Ts * psi / 360:  'angle d'amoráage pour créneaux

PRINT

PRINT "calcul signal en cours"

'calcul du signal

CalculXlin x(), 0, Tf * n / N1

I = 0:  'numéro de point

K = 0:  'numéro de période

WHILE I < n

    T = x(I) - K * Ts:        'initialisation du temps dans une période

    ON formesignal GOSUB CARRE, DENTDESCIE, SINUS, SINUSRED, CRENEAU, MLI, MODAMPL, MODFREQ, FSK, DIRAC, UN, SYNTHESE, SINECH, SINECH

    I = I - 1

    K = K + 1

WEND

SELECT CASE formesignal

    CASE 6

        FOR I = 0 TO n

            IF y(I, 0) < v * SIN(2 * pi# * x(I) / Tf) THEN y(I, 0) = -1 ELSE y(I, J) = 1

        NEXT I

    CASE 13

         FOR I = 0 TO n

            y(I, 0) = y(I, 0) * v * SIN(2 * pi# * x(I) / Tf)

         NEXT I

    CASE 14

        FOR I = 0 TO n

            IF y(I, 0) = 1 THEN

                y(I, 0) = y(I, 0) * v * SIN(2 * pi# * x(I) / Tf)

            ELSE

                y(I, 0) = y(I - 1, 0)

            END IF

       NEXT I

END SELECT

'le signal est maintenant dans le tableau y(i,0)

'FOR i = 0 TO N

'PRINT i, y(i, 0)

'    FOR h = 0 TO 2000: NEXT h

'NEXT i

EXIT SUB

CARRE:

WHILE T < Ts AND I <= n

    T = x(I) - K * Ts

    IF T < Ts * rapportcyclique THEN y(I, 0) = v ELSE y(I, 0) = 0

    I = I + 1

WEND

RETURN

DENTDESCIE:

WHILE T < Ts AND I <= n

    T = x(I) - K * Ts

    IF T < Ts / 4 THEN

        y(I, 0) = T * v / (Ts / 4)

    ELSEIF T >= Ts / 4 AND T < 3 * Ts / 4 THEN

        y(I, 0) = -(T - Ts / 4) * v / (Ts / 4) + v

    ELSE

        y(I, 0) = (T - 3 * Ts / 4) * v / (Ts / 4) - v

    END IF

    I = I + 1

WEND

RETURN

SINUS:

WHILE T < Ts AND I <= n

    T = x(I) - K * Ts

    y(I, 0) = v * SIN(2 * pi# * T / Ts)

    I = I + 1

WEND

RETURN

SINUSRED:

    WHILE T < Ts AND I <= n

        T = x(I) - K * Ts

        y(I, 0) = v * ABS(SIN(2 * pi# * T / Ts))

        I = I + 1

    WEND

RETURN

CRENEAU:

WHILE T < Ts AND I <= n

    T = x(I) - K * Ts

    IF T < tpsi THEN

        y(I, 0) = 0

    ELSEIF T >= tpsi AND T < Ts / 2 - tpsi THEN

        y(I, 0) = v

    ELSEIF T >= Ts / 2 - tpsi AND T < Ts / 2 + tpsi THEN

        y(I, 0) = 0

    ELSEIF T >= Ts / 2 + tpsi AND T < Ts - tpsi THEN

        y(I, 0) = -v

    ELSE

        y(I, 0) = 0

    END IF

    I = I + 1

WEND

RETURN

MLI:

    VV = v

    v = 1

    GOSUB DENTDESCIE

    v = VV

RETURN

MODAMPL:

    I = K

    y(I, 0) = (1 + v * SIN(2 * pi# * x(I) / Tf)) * SIN(2 * pi# * x(I) / Ts)

    I = I + 1

RETURN

MODFREQ:

    I = K

    y(I, 0) = COS(2 * pi# * x(I) / Ts + v * Tf * SIN(2 * pi# * x(I) / Tf))

    I = I + 1

RETURN

FSK:

WHILE I <= N1 / 2 - 1

    IF SIN(2 * pi# * x(I) / Ts) > 0 THEN y(I, 0) = v ELSE y(I, 0) = -v

    I = I + 1

WEND

WHILE I <= n

    IF SIN(pi# * x(I) / Ts) > 0 THEN y(I, 0) = v ELSE y(I, 0) = -v

    I = I + 1

WEND

RETURN

DIRAC:

y(I, 0) = v

I = I + 1

WHILE T < Ts AND I <= n

    T = x(I) - K * Ts

    I = I + 1

WEND

RETURN

UN:

I = K

y(I, 0) = v

I = I + 1

RETURN

SYNTHESE:

I = K

y(I, 0) = Vampli(0)

FOR h = 1 TO 10

    y(I, 0) = y(I, 0) + Vampli(h) * SIN(2 * pi# * x(I) * h / Ts + Phi(h))

NEXT h

I = I + 1

RETURN

SINECH:

VV = v

v = 1

GOSUB DIRAC

v = VV

RETURN

END SUB

SUB graph (x() AS SINGLE, Xax AS AXE, y() AS SINGLE, Yax AS AXE, aspect AS GRAPHE)

'trace le graphe des fonctions y = f(x,j) paramétrées en j

'tableaux commenáant à l'indice 0 (OPTION BASE 0)

'calcule automatiquement l'échelle et les unités si celles-ci sont nulles

'tableaux (option base 0)

n = UBOUND(y, 1)

m = UBOUND(y, 2)

DIM xgrad(10) AS SINGLE

DIM ygrad(10) AS SINGLE

PRINT "préparation du graphe..."

'Contrìles et vérifications... : contrìle des tableaux

IF n < 2 OR UBOUND(x) <> n OR x(n) - x(0) <= 0 THEN

    alerte "graphe impossible : mauvais ou absent tableau de valeurs en abscisse !"

    EXIT SUB

END IF

IF Xax.max < Xax.min OR Yax.max < Yax.min THEN

    alerte "graphe impossible : mauvaise échelle"

    EXIT SUB

END IF

'initialisation : calcul des extrema de la fonction

CalculExtrema y(), yminf, ymaxf, mini, maxi, yposmin

xminf = x(mini)

xmaxf = x(maxi)

I = 0

WHILE x(I) <= 0 AND I < n - 1

    I = I + 1

WEND

xposmin = x(I)

'Contrìles et vérifications pour axes logarithmiques

'si il n'existe pas de plus petite valeur strictement positive en abscisse et/ou en ordonnée, abandonner.

IF (Xax.linlog = -1 AND xposmin <= 0) OR (Yax.linlog = -1 AND yposmin >= 3.4E+38) THEN

    alerte "graphe impossible : toutes valeurs <= 0 sur axe(s) logarithmique(s) !"

    EXIT SUB

END IF

'si l'échelle, lorsqu'elle est précisée, n'est pas strictement positive, abandonner

IF (Xax.linlog = -1 AND NOT (Xax.min = 0 AND Xax.max = 0)) THEN

    IF NOT (Xax.min > 0 AND Xax.max > 0) THEN

        alerte "graphe impossible : échelle à borne <= 0 sur axe logarithmique en abscisse !"

        EXIT SUB

    END IF

END IF

IF (Yax.linlog = -1 AND NOT (Yax.min = 0 AND Yax.max = 0)) THEN

    IF (Yax.min > 0 AND Yax.max > 0) THEN

        alerte "graphe impossible : échelle à borne <= 0 sur axe logarithmique en ordonnée !"

        EXIT SUB

    END IF

END IF

'Valeurs par défaut

'si Xax.max = Xax.min = 0 (pas d'échelle précisée en abscisse), valeurs par défaut

IF Xax.max = 0 AND Xax.min = 0 THEN

    Xax.max = x(n)

    IF NOT Xax.linlog THEN

        Xax.min = x(0)

    ELSE

        Xax.min = xposmin

    END IF

END IF

'si pas d'échelle précisée en ordonnée, valeurs par défaut

IF Yax.max = 0 AND Yax.min = 0 THEN

    IF NOT Yax.linlog THEN

        IF ymaxf > yminf THEN

            Yax.max = ymaxf

            Yax.min = yminf

        ELSE

            SELECT CASE ymaxf

                CASE IS > 0

                    Yax.max = ymaxf

                    Yax.min = 0

                CASE IS < 0

                    Yax.max = 0

                    Yax.min = yminf

                CASE 0

                    Yax.max = 1

                    Yax.min = -1

            END SELECT

        END IF

    ELSE

        IF ymaxf > yposmin THEN

            Yax.max = ymaxf

            Yax.min = yposmin

        ELSE

            SELECT CASE ymaxf

                CASE IS > 1

                    Yax.max = ymaxf

                    Yax.min = 1

                CASE IS < 1

                    Yax.max = 1

                    Yax.min = ymaxf

                CASE 1

                    Yax.max = 10

                    Yax.min = .1

            END SELECT

       END IF

    END IF

END IF

' définition des paramètres de l'écran

SELECT CASE aspect.mode

    CASE 0

        alerte "graphe impossible : écran en mode texte !"

        EXIT SUB

    CASE 1

        lemax = 320: hemax = 200: nbcol = 40: nbligne = 25: nbcouleur = 1

    CASE 2

        lemax = 640: hemax = 200: nbcol = 80: nbligne = 25: nbcouleur = 1

    CASE 3

        lemax = 720: hemax = 348: nbcol = 80: nbligne = 25: nbcouleur = 1

    CASE 4

        lemax = 640: hemax = 400: nbcol = 80: nbligne = 25: nbcouleur = 16

    CASE 7, 13

        lemax = 320: hemax = 200: nbcol = 40: nbligne = 25: nbcouleur = 16

    CASE 8

        lemax = 640: hemax = 200: nbcol = 80: nbligne = 25: nbcouleur = 16

    CASE 9

        lemax = 640: hemax = 350: nbcol = 80: nbligne = 25: nbcouleur = 4

    CASE 10

        lemax = 640: hemax = 350: nbcol = 80: nbligne = 25: nbcouleur = 1

    CASE 11

        lemax = 640: hemax = 480: nbcol = 80: nbligne = 30: nbcouleur = 1

    CASE 12

        lemax = 640: hemax = 480: nbcol = 80: nbligne = 30: nbcouleur = 16

    CASE ELSE

        alerte "graphe impossible : mode graphique inconnu !"

        EXIT SUB

END SELECT

'paramètres d'impression des nombres

largecaract = lemax / nbcol:  'largeur en pixels d'un caractère

hautcaract = hemax / nbligne: 'hauteur en pixels d'un caractère

'définition de la fenàtre graphique physique

CLS 0

SCREEN aspect.mode

margeG = formatreel * largecaract

margeD = 3 * largecaract

margeH = 2.5 * hautcaract

margeB = 4.5 * hautcaract

VIEW (margeG, margeH)-(lemax - margeD, hemax - margeB)

'définition de la fenàtre graphique logique

wxmin = wlog(Xax.min, Xax.linlog)

wxmax = wlog(Xax.max, Xax.linlog)

wymin = wlog(Yax.min, Yax.linlog)

wymax = wlog(Yax.max, Yax.linlog)

WINDOW (wxmin, wymax)-(wxmax, wymin)

grisref = 12

'détermination de l'origine des axes

IF Xax.min < 0 AND Xax.max >= 0 THEN xorig = 0 ELSE xorig = Xax.min

IF Yax.min < 0 AND Yax.max >= 0 THEN yorig = 0 ELSE yorig = Yax.min

wxorig = wlog(xorig, Xax.linlog)

wyorig = wlog(yorig, Yax.linlog)

'tracé des axes et du cadre

LINE (wxmin, wyorig)-(wxmax, wyorig), grisref

LINE (wxorig, wymin)-(wxorig, wymax), grisref

LINE (wxmin, wymax)-(wxmax, wymin), grisref, B

IF nbcouleur > 1 THEN COLOR 10

'tracé des graduations en x

ylonggrad = (wymax - wymin) * 6 / (hemax - margeH - margeB)

calculgrad Xax, xgrad()

FOR I = 0 TO UBOUND(xgrad)

    wx = wlog(xgrad(I), Xax.linlog)

    IF aspect.grille THEN LINE (wx, wymax)-(wx, wymin), , , &H8888

    LINE (wx, wyorig - ylonggrad)-(wx, wyorig + ylonggrad), grisref

    LINE (wx, wymax)-(wx, wymax - ylonggrad), grisref

    LINE (wx, wymin + ylonggrad)-(wx, wymin), grisref

    xreel$ = affichegrad$((xgrad(I)), Xax)

    nocol = INT((POINT(0) + margeG) / largecaract) + 1 - LEN(xreel$) / 2

    noligne = INT((POINT(1) + margeH) / hautcaract) + 2

    IF nocol < nbcol THEN

        LOCATE noligne, nocol

        PRINT xreel$;

    END IF

NEXT I

'tracé des graduations en y

xlonggrad = (wxmax - wxmin) * 6 / (lemax - margeG - margeD)

calculgrad Yax, ygrad()

FOR I = 0 TO UBOUND(ygrad)

    wy = wlog(ygrad(I), Yax.linlog)

    IF aspect.grille THEN LINE (wxmin, wy)-(wxmax, wy), , , &H8888

    LINE (wxorig - xlonggrad, wy)-(wxorig + xlonggrad, wy), grisref

    LINE (wxmax, wy)-(wxmax - xlonggrad, wy), grisref

    LINE (wxmin + xlonggrad, wy)-(wxmin, wy), grisref

    yreel$ = affichegrad$((ygrad(I)), Yax)

    nocol = INT((POINT(0) + margeG) / largecaract) - LEN(yreel$)

    noligne = INT((POINT(1) + margeH) / hautcaract) + 1

    IF noligne >= 2 THEN

        LOCATE noligne, nocol

        PRINT yreel$;

    END IF

NEXT I

'repérage des courbes par la couleur

IF nbcouleur > 1 THEN

    LOCATE 5, 1

    FOR J = 1 TO nbcouleur - 1

        gris = nbcouleur - J:    'couleurs des courbes

        IF gris = 14 THEN gris = 7:  'le jaune n'est pas reproductible sur certaines imprimantes !

        COLOR gris

        PRINT "€"; J

    NEXT J

END IF

'tracé des courbes

FOR J = 0 TO m

    gris = 15 - J MOD 15:   'couleurs des courbes

    IF gris = 14 THEN gris = 7

    PSET (wlog(x(0), Xax.linlog), wlog(y(0, J), Yax.linlog)), gris

    FOR I = indexmin TO n

        wx = wlog(x(I), Xax.linlog)

        wy = wlog(y(I, J), Yax.linlog)

        SELECT CASE aspect.ligne

            CASE 1, 3

                LINE -(wx, wy), gris

            CASE 2

                LINE (wx, wyorig)-(wx, wy), gris

            CASE 0

                wancy = POINT(3)

                LINE -(wx, wancy), gris

                LINE -(wx, wy), gris

        END SELECT

        SELECT CASE aspect.forme

            CASE 1

                LINE (wx, wy - ylonggrad)-(wx, wy + ylonggrad), gris

                LINE (wx - xlonggrad, wy)-(wx + xlonggrad, wy), gris

            CASE 2

                CIRCLE (wx, wy), ylonggrad / 3, 1

                PAINT (wx, wy), gris, 1

        END SELECT

        PSET (wx, wy), gris

    NEXT I

NEXT J

'afficher les valeurs des extrema et sortir

IF nbcouleur > 1 THEN COLOR grisref

VIEW PRINT nbligne - 2 TO nbligne

PRINT "mini : x ="; xminf; TAB(formatreel + 10); "y ="; yminf:

PRINT "maxi : x ="; xmaxf; TAB(formatreel + 10); "y ="; ymaxf:

PRINT "nombre de points :"; n + 1;

PRINT TAB(nbcol - 40); "appuyer sur une touche pour continuer";

IF nbcol = 80 THEN

    LOCATE nbligne - 2, 3 * nbcol / 4

    PRINT "Date  : "; DATE$

    LOCATE nbligne - 1, 3 * nbcol / 4

    PRINT "Heure : "; TIME$

END IF

DO: LOOP WHILE INKEY$ = ""

IF nbcouleur > 1 THEN COLOR 15

VIEW PRINT

END SUB

DEFSNG A-D

SUB imprim (x() AS SINGLE, y() AS SINGLE)

n = UBOUND(y, 1)

m = UBOUND(y, 2)

FOR J = 0 TO m

        FOR I = 0 TO n

               LPRINT x(I), y(I, J)

        NEXT I

NEXT J

END SUB

DEFSTR A-D

SUB spectreampli (x() AS SINGLE, y() AS SINGLE)

'calcule le spectre d'amplitude du signal y(i,0) dans le tableau y(i,0)

'la largeur du spectre est égale à rmax=(N+1)/2, rang de l'harmonique de fréquence la plus haute

n = UBOUND(y)

DIM yRe(n) AS SINGLE, yIm(n) AS SINGLE

FOR I = 0 TO n

    yRe(I) = y(I, 0)

    yIm(I) = 0

NEXT I

FFT yRe(), yIm(), "F"

rmin = -(n + 1) / 2 + 1

rmax = (n + 1) / 2

CalculXlin x(), rmin, rmax

FOR I = 0 TO (n + 1) / 2 - 2

    y(I, 0) = 0

NEXT I

FOR I = (n + 1) / 2 - 1 TO n

    y(I, 0) = 2 * SQR(yRe(I) ^ 2 + yIm(I) ^ 2)

NEXT I

y((n + 1) / 2 - 1, 0) = y((n + 1) / 2 - 1, 0) / 2

END SUB

DEFSNG A-D

FUNCTION wlog (s AS SINGLE, flaglinlog AS INTEGER)

IF flaglinlog THEN

    IF s > 0 THEN wlog = LOG(s) ELSE wlog = -100

ELSE wlog = s

END IF

END FUNCTION


